COMPUTER PROGRAMS

FOR

LOCATING THE MOON

DIVISION OF VARIAN
301 Industrial Way
San Carlos, California

AS-49-6

4145

COMPUTER PROGRAMS FOR LOCATING THE MOON

In two previous issues of the EIMAC EME notes, methods were described for determining antenna aiming data to "hit" the moon. In Bulletin AS-49, there was a reprint of an article by Don Lund, WAØIQN, an article by Victor Michael, W3SDZ, and some letters to the Technical Editor of QST from Alan Goodacre, VE3BZS, all discussing antenna aiming. Also, in Bulletin AS-49-1 more information on aiming data was presented. Bulletin AS-49-1 stressed the use of the Nautical Almanac and the Hydrographic Office publication HO-214. It is possible to determine the moon's position by means of these two publications without extensive mathematics. However, as one becomes more involved in the EME project, the habits of the moon become interesting. Also, more sophisticated methods of tracking the moon become attractive.

For those persons having access to the hand-held scientific calculators, new techniques become available. For those amateurs having access to a computer, either punch card, or an active terminal, another interesting hobby of writing and modifying programs develops.

In this issue of the EME notes, four programs are presented which may be of interest. One is in BASIC, two are in TYMSHARE SUPERBASIC and one in TYMSHARE SUPERFORTRAN. As is usually the case, these programs will probably need to be modified to run on the machine available to the individual amateur.

It should be emphasized that a computer is not necessary to have successful moonbounce contacts. A computer is just another interesting tool to work with which can add to the fun.

PROGRAM I

LANGUAGE: BASIC

This program will run on almost all machines accepting BASIC. The operator must put in data from the Nautical Almanac for the year, month, day and hours that the elevation and azimuth is desired. Also, the latitude and longitude of the geographic location in question must be entered. Statement 20 allows the operator to put in the time interval, in minutes, between each calculation.

The origin of this program is not known.

PROGRAM 2

LANGUGAGE: TYMSHARE SUPERBASIC

The program is a modified version of Program One. The output format is slightly different. Also, limits can be put on the azimuth and elevation angles at the operator's discretion. The program requires latitude and longitude and time interval inputs. Also, Nautical Almanac data for the year, month, day and hours must be put into the data statements. Statement 610 allows the operator to put the data in for identification. The computer

does not use this statement in the calculations. The computer will pause and let you type in the date. When inserting the date, do not use commas for separating the month, day and year. If you do, the computer will become confused. When the carriage return is depressed the computer will carry on.

PROGRAM 3

LANGUAGE: TYMSHARE SUPERBASIC

This program was originally written in BASIC by Lance Collister, WA3GPL. However, the BASIC used on the General Electric terminal was sufficiently different from the TYMSHARE SUPERBASIC, that the program would not run. A few modifications were made to allow the TYMSHARE machine to run. What is presented here will not run on the GE computer. Perhaps those who attempt to use this program will have to make further changes for their individual cases.

The program is very useful because no almanac data is required. The program allows the calculation of the moon position by comparing back to the position of the moon at the start of the Julian calendar. The operator need put in only the latitude, longitude, time interval, month, day, year and hours between which aiming data is required. If rising, or setting, moon calculations are required, the computer determines the time and will execute the printout requested.

PROGRAM 4

LANGUAGE: TYMSHARE SUPERFORTRAN

This program was supplied by Louis Anciaux, WB6NMT. As received, the program was on tape and written in Fortran IV. Again, the program was modified to run with TYMSHARE SUPERFORTRAN. This program requires Nautical Almanac data. However, the GHA and DECLINATION for 0000, 0600, 1200, 1800, and 2400 hours GMT is all that is required. The program will interpolate for all other times. Of course the latitude, longitude and time interval for the calculation must be typed in.

PROGRAM I
LANGUAGE: BASIC

```
10  DATA 37,23.5,122,10.5
20  DATA 5
100  REM THIS PROGRAM CALCULATES MOON AIMING DATA FROM A SPECIFIED
110  REM QTH AT SELECTED TIME INTERVALS DURING THE HOUR.
120  REM
130  REM DATA REQUIREMENTS ARE AS FOLLOWS:
140  REM
150  REM LINE 10 - LATITUDE DEGREES, LATITUDE MINUTES,
160  REM           LONGITUDE DEGREES, LONGIDUDE MINUTES
170  REM
180  REM LINE 20 - TIME INTERVAL FOR CALCULATIONS IN MINUTES
190  REM
200  REM LINE 1050--ALMANAC DATA CONSISTING OF THE TIME, GHA DEGREES
210  REM           GHA MINUTES, DECLINATION DEGREES AND DECLINATION
220  REM           MINUTES
230  REM
240  REM           DATA MUST BE GIVEN FOR BOTH THE BEGINNING AND
250  REM           END OF THE HOUR.  THAT IS, IF AIMING DATA IS
260  REM           DESIRED FOR HOURS BEGINNING 0900,1000,AND 1100
270  REM           YOU MUST SHOW DATA FOR 0900,1000,1100 AND 1200.
280  REM
290  REM TO INDICATE THE END OF THE HOURLY DATA  INCLUDE A SINGLE
300  REM DATA LINE WITH A ONE, A TWO FOR FINISHED MONTH
305  REM WITH THIS METHOD A MONTH AT A TIME CAN BE DONE
310  REM
320  LET R=1.74533E-02
330  LET P1=180*R
340  LET P3=360*R
350  READ L3,L4
360  LET L1=(L3+L4/60)*R
370  READ L3,L4
380  LET L2=(L3+L4/60)*R
390  LET L3=SIN(L1)
400  LET L4=COS(L1)
405  READ J1
406  LET J=60/J1
410  PRINT
420  PRINT
430  PRINT
440  PRINT "MOON AIMING DATA SPECIALLY PREPARED FOR K6MYC"
450  PRINT
460  PRINT
470  PRINT "TIME","AZIMUTH","ELEVATION"
480  PRINT
510  READ B
530  READ B1,B2,B3,B4
540  LET G1=(B1+B2/60)*R
550  LET D1=(B3+B4/60)*R
560  READ B5
575  IF B5=1 THEN 450
 >
```

```
576  IF B5=2 THEN 2000
577  IF B5=3 THEN 480
580  READ B1,B2,B3,B4
590  LET G2=(B1+B2/60)*R
600  LET D2=(B3+B4/60)*R
610  LET G0=(G2-G1)/J
620  IF G0>0 THEN 640
630  LET G0=(G2+P3-G1)/J
640  LET D0=(D2-D1)/J
650  LET G=G1
660  LET D=D1
670  FOR I=0 TO 59 STEP J1
680  LET D3=SIN(D)
690  LET D4=COS(D)
700  LET H=L2-G
710  IF H-P1<0 THEN 750
720  IF H-P1=0 THEN 770
730  LET H=P3-H
740  GOTO 770
750  IF H+P1 >= 0 THEN 770
760  LET H=P3+H
770  LET A=0
780  LET E=0
790  LET E3=L3*D3+L4*D4*COS(H)
800  IF E3<0 THEN 940
810  LET E4=SQR(1-E3↑2)
820  LET A3=SIN(H)*D4/E4
830  LET A4=SQR(1-A3↑2)
840  LET A=ATN(A3/A4)
850  LET E0=E3/E4-4/(240*E4)
860  IF E0 <= 0 THEN 950
870  LET E=ATN(E0)/R
880  LET E=INT(10*E+.5)/10
890  IF L3*E3-D3>0 THEN 930
900  IF A >= 0 THEN 940
910  LET A=P3+A
920  GOTO 940
930  LET A=P1-A
940  LET A=INT(10*A/R+.5)/10
945  IF E=0 THEN 960
950  PRINT B+I,A,E
960  LET G=G+G0
970  LET D=D+D0
980  NEXT I
990  LET G1=G2
1000  LET D1=D2
1010  LET B=B5
1020  GOTO 560
2000  END
  >
```

PROGRAM 2
LANGUAGE: TYMSHARE SUPERBASIC

```
100  DATA 37,34,122,18
110  DATA 15
120  REM THIS PROGRAM CALCULATES MOON AIMING DATA FROM A SPECIFIED
130  REM QTH AT SELECTED TIME INTERVALS DURING THE HOUR.
140  REM
150  REM DATA REQUIREMENTS ARE AS FOLLOWS:
160  REM
170  REM LINE 100 -LATITUDE DEGREES, LATITUDE MINUTES,
180  REM           LONGITUDE DEGREES, LONGITUDE MINUTES
190  REM
200  REM LINE 110 -TIME INTERVAL FOR CALCULATIONS IN MINUTES
210  REM
220  REM LINE 1270--ALMANAC DATA CONSISTING OF THE TIME, GHA DEGREES
230  REM           GHA MINUTES, DECLINATION DEGREES AND DECLINATION
240  REM           MINUTES
250  REM
260  REM           DATA MUST BE GIVEN FOR BOTH THE BEGINNING AND
270  REM           END OF THE HOUR.  THAT IS, IF AIMING DATA IS
280  REM           DESIRED FOR HOURS BEGINNING 0900,1000,AND 1100
290  REM           YOU MUST SHOW DATA FOR 0900,1000,1100 AND 1200.
300  REM
310  REM TO INDICATE THE END OF THE HOURLY DATA  INCLUDE A SINGLE
320  REM DATA LINE WITH A ONE, A TWO FOR FINISHED MONTH
330  REM WITH THIS METHOD A MONTH AT A TIME CAN BE DONE
340  REM
350  LET R=1.74533E-02
360  LET P1=180*R
370  LET P3=360*R
380  READ L3,L4
390  LET L1=(L3+L4/60)*R
400  READ L3,L4
410  LET L2=(L3+L4/60)*R
420  LET L3=SIN(L1)
430  LET L4=COS(L1)
440  READ J1
450  LET J=60/J1
460  PRINT
470  PRINT
480  PRINT
490  PRINT "MOON AIMING DATA FOR W6PO."
>
```

```
500  PRINT
510  PRINT "MINIMUM AZIMUTH":
520  INPUT Z1
530  PRINT "MAXIMUM AZIMUTH":
540  INPUT Z2
550  PRINT "MINIMUM ELEVATION":
560  INPUT M
570  PRINT "MAXIMUM ELEVATION":
580  INPUT N
590  PRINT
600  PRINT
610  PRINT "DATE":
620  INPUT X
630  PRINT
640  PRINT
650  PRINT "GMT","AZIMUTH","ELEVATION"
660  PRINT
670  READ B
680  READ B1,B2,B3,B4
690  LET G1=(B1+B2/60)*R
700  LET D1=(B3+B4/60)*R
710  READ B5
720  IF B5=1 THEN 590
730  IF B5=2 THEN 2000
740  IF B5=3 THEN 660
750  READ B1,B2,B3,B4
760  LET G2=(B1+B2/60)*R
770  LET D2=(B3+B4/60)*R
780  LET G0=(G2-G1)/J
790  IF G0>0 THEN 810
800  LET G0=(G2+P3-G1)/J
810  LET D0=(D2-D1)/J
820  LET G=G1
830  LET D=D1
840  FOR I=0 TO 59 STEP J1
850  LET D3=SIN(D)
860  LET D4=COS(D)
870  LET H=L2-G
880  IF H-P1<0 THEN 920
890  IF H-P1=0 THEN 940
  >
```

```
900  LET H=P3-H
910  GOTO 940
920  IF H+P1 >= 0 THEN 940
930  LET H=P3+H
940  LET A=0
950  LET E=0
960  LET E3=L3*D3+L4*D4*COS(H)
970  IF E3<0 THEN 1110
980  LET E4=SQR(1-E3↑2)
990  LET A3=SIN(H)*D4/E4
1000  LET A4=SQR(1-A3↑2)
1010  LET A=ATN(A3/A4)
1020  LET E0=E3/E4-4/(240*E4)
1030  IF E0 <= 0 THEN 1170
1040  LET E=ATN(E0)/R
1050  LET E=INT(10*E+.5)/10
1060  IF L3*E3-D3>0 THEN 1100
1070  IF A >= 0 THEN 1110
1080  LET A=P3+A
1090  GOTO 1110
1100  LET A=P1-A
1110  LET A=INT(10*A/R+.5)/10
1120  IF E=0 THEN 1200
1130  IF A>Z1 THEN 1140 ELSE 1200
1140  IF A<Z2 THEN 1150 ELSE 1200
1150  IF E>M THEN 1160 ELSE 1200
1160  IF E<N THEN 1170 ELSE 1200
1170  PRINT IN FORM "4D":(B+I)
1180  PRINT IN IMAGE "           %%%.%          %%.%":A,E
1190  IF I=45 THEN PRINT
1200  LET G=G+G0
1210  LET D=D+D0
1220  NEXT I
1230  LET G1=G2
1240  LET D1=D2
1250  LET B=B5
1260  GOTO 710
2000  END
 >
```

PROGRAM 3
LANGUAGE: TYMSHARE BUPERBASIC

```
100!     THIS PROGRAM IS DESIGNED TO CALCULATE THE AZIMUTH AND
110!ELEVATION OF THE MOON.
120!
130!     THE REQUIRED INPUT IS THE GMT MONTH,DAY,AND YEAR FOR
140!WHICH THE MOON'S AZIMUTH AND ELEVATION ARE DESIRED.  IN
150!ADDITION, IF PRINTOUT IS DESIRED NOT ONLY WHEN THE MOON IS NEAR
160!THE HORIZON, THE TIME INTERVAL OVER WHICH THE POSITION IS DESIRED
170!MUST BE ENTERED.
180!
190!     THE COMPLETE INPUT FORMAT FOR THE ABOVE INFORMATION IS:
200!
210!                  MM,DD,YYYY,TTTT,TTTT
220!
230!     IF YOU ELECT TO HAVE PRINTOUT OCCUR ONLY WHEN THE MOON
240!NEAR THE HORIZON, YOU WILL BE ASKED TO INPUT THE MAXIMUM ELEVATION
250!FOR WHICH YOU WANT PRINTOUT. IN THIS CASE, YOU WILL NOT BE ASKED
260!TO SUPPLY TIMES FOR THE TIME INTERVALS.
270!
280!     PRINTOUT IS SUSPENDED WHENEVER THE ELEVATION OF THE MOON IS
290!NEGATIVE.
300!
310!     YOU CAN ASK FOR DATA FOR UP TO 25 DAYS AT ONCE:SIMPLY
320!TYPE IN THE INFORMATION FOR ONE DAY (AS IN THE REQUIRED FORMAT
330!FOR YOUR PARTICULAR NEEDS) EACH TIME THE COMPUTER PRINTS A ? .
340!WHEN YOU HAVE ENTERED ALL THE DATA YOU WISH,FOLLOW THE LAST ? BY AN
350!INPUT OF ZEROES SEPARATED BY COMMAS-JUST AS IN THE FORMAT YOU WERE
360!USING TO ENTER THE DATA.
370!
380!     ALL DATES AND TIMES USED IN THIS PROGRAM ARE IN GREENWICH MEAN
390!TIME.  IN ADDITION, ALL TIMES ARE IN THE 0000 TO 2400 HOUR SYSTEM.
400!
410!     THE BASIC USED IN THIS PROGRAM HAS BEEN KEPT AS ELEMENTARY AS
420!POSSIBLE TO FACILITATE ITS USE ON OTHER TIME SHARING SYSTEMS. I
430!WISH TO ACKNOWLEDGE THE GENEROUS ASSISTANCE OF THOMAS AKE OF THE
440!WARNER-SWASEY OBSERVATORY, WITHOUT WHOM THE VITAL EQUATIONS FOR
450!COMPUTING THE MOON'S POSITION WOULD NOT HAVE BEEN AVAILABLE.
460!
470!                                        LANCE COLLISTER
480!                                        WA1JXN/WA3GPL
490!                                        CLEVELAND, OHIO
500!                                        MARCH, 1974
510!
520!*************************************************************
530 DIM F(25),V(25),Y(25),Q(25),S(25)
540 DEF FNA(X)=INT(X*D5*10+.5)/10
550 DEF FNB(X)=(X-INT(X))*P5
560 LET P5=2.0000000000*3.1415926535
570 LET D5=360.0000000000/P5 !CONVERSION TO DEGREES
580 LET R5=P5/360.0000000000 !CONVERSION TO RADIANS
583 STRING Z8
584 PRINT
585 PRINT "WHAT ARE THE CALL LETTERS OF THE STATION":
586 INPUT Z8
590 PRINT "WHAT IS THE LATITUDE IN DEGREES,MINUTES":
 >
```

```
600 INPUT L5,U5
610 PRINT "WHAT IS THE LONGITUDE IN DEGREES,MINUTES":
620 INPUT L6,U6
630 LET L5=(L5+U5/60)*R5
640 LET L6=(L6+U6/60)*R5
650 PRINT "WHAT IS THE DESIRED PRINTING INCREMENT IN MINUTES":
660 INPUT I
670 PRINT "DO YOU ONLY WANT PRINTOUT WHEN THE MOON"
680 PRINT "IS NEAR THE HORIZON":
690 INPUT B$
700 IF B$="YES" THEN 730
710 LET I6=100
720 GOTO 800
730 PRINT"BELOW WHAT ELEVATION IN DEGREES DO YOU WANT PRINTOUT TO OCCUR?
740 INPUT I6
750 PRINT "WHAT ARE THE GMT MONTH,DAY,YEAR DESIRED":
760 FOR N=1 TO 25
770 INPUT F(N),V(N),Y(N)
780 IF F(N)=0 THEN 860
785 NEXT N
790 GOTO 760
800 PRINT "WHAT ARE THE GMT MONTH,DAY,YEAR,TIME INTERVAL(BEGINNING,"
810 PRINT "ENDING) DESIRED":
820 FOR N=1 TO 25
830 INPUT F(N),V(N),Y(N),Q(N),S(N)
840 IF F(N)=0 THEN 860
845 NEXT N
850 GOTO 820
860 LET N5=N-1
870 FOR N=1 TO N5
880 IF B$="YES" THEN 900
890 GOTO 930
900 LET E1=2400
910 LET B=0
920 GOTO 950
930 LET E1=S(N)
940 LET B=Q(N)
950 LET M=F(N)
960 LET D=V(N)
970 LET Y=Y(N)
980 LET Y1=Y-(INT(Y/100)*100)
990 PRINT
1000 PRINT
1010 PRINT "POSITION OF MOON ON ":M:"/":D:"/":Y1:"GMT"
1020 PRINT
1030 PRINT "GMT","AZ","EL","GHA","DEC"
1040 PRINT "---","--","--","---","---"
1050 PRINT
1060 LET I1=2
1070 !HERE BEGINS CALCULATION OF JULIAN DATE
1080 IF M>=3 THEN 1160
1090 IF INT((Y-1853)/4)<11 THEN 1120
1100 LET C1=-1
1110 GOTO 1130
 >
```

```
1120 LET C1=0
1130 LET J1=365*(Y-1853)+D+30*(M+9)+INT((M+10)/2)
1140 LET J2=INT((Y-1853)/4)+1+C1
1150 GOTO 1270
1160 IF INT((Y-1852)/4)<11 THEN 1190
1170 LET C1=-1
1180 GOTO 1200
1190 LET C1=0
1200 IF M=9 THEN 1240
1210 IF M=11 THEN 1240
1220 LET C2=0
1230 GOTO 1250
1240 LET C2=1
1250 LET J1=365*(Y-1852)+D+30*(M-3)+INT((M-2)/2)
1260 LET J2=INT((Y-1852)/4)+C1+C2
1270 LET J=J1+J2 !(JULIAN DATE-2397547.5) FOR 0 HOURS GMT
1280 LET T1=J-17472.5
1290 LET D9=(B-INT(B/100)*100)+INT(B/100)*60
1300 LET D6=(E1-INT(E1/100)*100)+INT(E1/100)*60
1310 LET D7=D9-D6
1320 LET D8=D7-I
1330 IF D7>0 THEN 1350
1340 GOTO 1380
1350 IF D8>=0 THEN 2220
1360 LET B=E1
1370 !CALCULATION OF LATITUDE AND LONGITUDE OF MOON
1380 LET T=(B-INT(B/100)*100)/1440+INT(B/100)/24
1390 LET T5=T1+T
1400 LET K1=FNB(.751213+.036601102*T5)
1410 LET K2=FNB(.822513+.0362916457*T5)
1420 LET K3=FNB(.995766+.00273777852*T5)
1430 LET K4=FNB(.974271+.0338631922*T5)
1440 LET K5=FNB(.0312525+.0367481957*T5)
1450 LET L8=K1+.658*R5*SIN(2*K4)+6.289*R5*SIN(K2)
1460 LET L8=L8-1.274*R5*SIN(K2-2*K4)-.186*R5*SIN(K3)
1470 LET L8=L8+.214*R5*SIN(2*K2)-.114*R5*SIN(2*K5)
1480 LET L8=L8-.059*R5*SIN(2*K2-2*K4)-.057*R5*SIN(K2+K3-2*K4)
1490 LET K6=K5+.6593*R5*SIN(2*K4)+6.2303*R5*SIN(K2)-1.272*R5*SIN(K2-2*K4
1500 LET L7=5.144*R5*SIN(K6)-.146*R5*SIN(K5-2*K4)
1510 !CALCULATION OF  RIGHT ASCENSION (A=R1) AND DECLINATION (D1)
1520 LET D1=COS(L7)*SIN(L8)*.397821+SIN(L7)*.917463
1530 LET D1=ATN(D1/(SQR(1-D1↑2)))
1540 LET A2=COS(L7)*COS(L8)/COS(D1)
1550 LET A1=(COS(L7)*SIN(L8)*.917463-SIN(L7)*.397821)/COS(D1)
1560 LET A=ATN(A1/A2)
1570 GOSUB 1860
1580 LET R1=A
1590 LET L1=.065709822*T1
1600 LET L=T*24*1.002738+6.646055+(L1-INT(L1/24)*24)
1610 LET L=(L-INT(L/24)*24)
1620! CALCULATION OF GREENWICH HOUR ANGLE, G, FROM LOCAL SIDEREAL TIME
1630 LET G=(L/24)*P5-R1
1640 IF G<P5 THEN 1670
1650 G=G-P5
>
```

```
1660 GOTO 1710
1670 IF G<0 THEN 1690
1680 GOTO 1710
1690 G=G+P5
1700! CALCULATION OF YOUR LOCAL HOUR ANGLE,H, FROM GHA
1710 LET H=L6-G
1720! CALCULATION OF ELEVATION,E,OF OBJECT
1730 LET E3=COS(L5)*COS(H)*COS(D1)+SIN(D1)*SIN(L5)     !SIN OF ELEVATION
1740 LET E2=SQR(1-(E3*E3))
1750 LET E=ATN(E3/E2)
1760 IF E<0 THEN 2170
1770 IF E>I6*R5 THEN 2170
1780! CALCULATION OF AZIMUTH,A,OF OBJECT
1790 LET A2=SIN(D1)/(COS(L5)*COS(E))
1800 LET A2=A2-(SIN(L5)/COS(L5))*(SIN(E)/COS(E))
1810 LET A1=SIN(L5)*SIN(D1)+COS(L5)*COS(D1)*COS(H)
1820 LET A1=(SIN(H)*COS(D1))/(SQR(1-A1↑2))
1830 LET A=ATN(A1/A2)      !AZIMUTH=ARCTAN( SIN OF AZ/ COS OF AZ)
1840 GOSUB 1860
1850 GOTO 2020
1860! REMOVAL OF AMBIGUITIES INCURRED WITH ATN FUNCTION
1870 IF A=0 THEN 1890
1880 GOTO 1930
1890 IF A2<0 THEN 1910
1900 GOTO 2010
1910 LET A=P5/2
1920 GOTO 2010
1930 IF A>0 THEN 1990
1940 IF A2<0 THEN 1970
1950 LET A=P5+A
1960 GOTO 2010
1970 LET A=P5+(A-P5/2)
1980 GOTO 2010
1990 IF A2>=0 THEN 2010
2000 LET A=A+P5/2
2010 RETURN
2020 IF (T-I1)>(2*I)/1440 THEN 2040
2030 GOTO 2145
2040 PRINT
2145 PRINT IN FORM "4D":(INT(B+.5))
2150 PRINT IN IMAGE "           %%%.%           %%.%
%%%.%          %%%.%":FNA(A),FNA(E),FNA(G),FNA(D1)
2160 LET I1=T
2170 LET B=B+I
2180 LET Z=(B-INT(B/100)*100)-60
2190 IF Z<0  THEN 1290
2200 LET B=INT(B/100)*100+100+Z
2210 GOTO 1290
2220 NEXT N
2230 LET N=0
2240 PRINT
2250 PRINT
2260 PRINT
2270 PRINT"DO YOU WANT MORE INFORMATION":
2280 INPUT D$
2290 IF D$="YES" THEN 540
2300 END
>
```

PROGRAM 4
LANGUAGE: TYMSHARE SUPERFORTRAN

```
100           C: PROGRAM CALCULATES AZ AND EL FROM GHA AND DEC...ANCIAUX
101           DIMENSION X(2,6)
102           STRING OBSER(8),DATE(15)
103       100 FORMAT(32H ENTER LATITUDE IN DEG  EG 056.3/)
104       101 FORMAT(F5.1)
105       102 FORMAT(33H ENTER LONGITUDE IN DEG EG 256.3/)
106       103 FORMAT(31H ENTER INTERVAL JUMP IN MINUTES/)
107       104 FORMAT(I2)
108       105 FORMAT(11H ENTER DATE/)
109       106 FORMAT(5A4)
110       107 FORMAT(16H ENTER 0000Z GHA/)
111       108 FORMAT(16H ENTER 0000Z DEC/)
112       109 FORMAT(16H ENTER 0600Z GHA/)
113       110 FORMAT(16H ENTER 0600Z DEC/)
114       111 FORMAT(16H ENTER 1200Z GHA/)
115       112 FORMAT(16H ENTER 1200Z DEC/)
116       113 FORMAT(16H ENTER 1800Z GHA/)
117       114 FORMAT(16H ENTER 1800Z DEC/)
118       115 FORMAT(16H ENTER 2400Z GHA/)
119       116 FORMAT(16H ENTER 2400Z DEC/)
120       117 FORMAT(15H ENTER OBSERVER/)
121       118 FORMAT(10A3)
122       119 FORMAT(16H AZ-EL DATA FOR , A6)
123       120 FORMAT(4H GMT,5X,3HGHA,6X,3HDEC,7X,2HAZ,7X,2HEL)
124       121 FORMAT(25H INPUT FOR GHA IS INVALID/)
125       122 FORMAT(37H FURTHER CALCS DESIRED THIS OBSERVER?)
126       123 FORMAT(23H TYPE 1 IF YES, 0 IF NO/)
127       124 FORMAT(I1)
128       125 FORMAT(I5,4(4X,F5.1))
129       126 FORMAT(39H FURTHER CALCS DESIRED FOR DIFF OBSVRS?)
130       127 FORMAT(6H  LONG,3X,F5.1,5X,3HLAT,3X,F5.1/)
131       128 FORMAT(5H     /)
132       129 FORMAT(A15)
133       200 WRITE(1,100)
134           ACCEPT BLAT
135           WRITE(1,102)
136           ACCEPT BLONG
137           WRITE(1,117)
138           ACCEPT OBSER
139       201 WRITE(1,103)
140           ACCEPT INTER
141           WRITE(1,105)
142           ACCEPT DATE
143           WRITE(1,107)
144           ACCEPT X(1,1)
145           WRITE(1,108)
146           ACCEPT X(2,1)
147           WRITE(1,109)
148           ACCEPT X(1,2)
149           WRITE(1,110)
>
```

```
150             ACCEPT X(2,2)
151             WRITE(1,111)
152             ACCEPT X(1,3)
153             WRITE(1,112)
154             ACCEPT X(2,3)
155             WRITE(1,113)
156             ACCEPT X(1,4)
157             WRITE(1,114)
158             ACCEPT X(2,4)
159             WRITE(1,115)
160             ACCEPT X(1,5)
161             WRITE(1,116)
162             ACCEPT X(2,5)
163             ITIME = 0
164             HOUR = 0
165             JTIME = 0
166             M = 1
167             JB = 1
168             JC = 1
169             JD = 1
170             JE = 1
171             HTEST = 600
172             PI = 3.1415926
173             CONVT = PI/180.0
174             CONVTR = 1.0/CONVT
175             CLAT = BLAT * CONVT
176             CSINL = SIN(CLAT)
177             CCOSL = COS(CLAT)
178             CTANL = CSINL/CCOSL
179             R1 = BLONG + 180.0
180             R2 = BLONG - 180.0
181             IF(R1.GE.360.0.OR.R2.LT.0.0)GO TO 202
182             GO TO 203
183     202     IF(R1 - 360.0)205,204,204
184     204     R1 = R1 - 360.0
185     205     IF(R2)206,203,203
186     206     R2 = R2 + 360.0
187             GO TO 225
188             C: ABOVE TESTS AND SETS R1 AND R2 TO UNDER 360 DEG
189     203     JE = 1
190     224     ITIME = ITIME + INTER
191             IF(ITIME - 60)210,208,208
192     208     ITIME = ITIME - 60
193             HOUR = HOUR + 100
194             IF(HOUR - HTEST)210,209,209
195     209     HTEST = HTEST + 600
196             M = M+1
197             JC = 1
198             JD = 1
199     210     JTIME = HOUR + ITIME
>
```

```
200             IF(JTIME - 2400)225,225,300
201             C: INCREMENTS TIME AND SETS TO HRS AND MINS, TESTS FOR 6 HR
202             C: M INCREMENTS MATRIX ELEMENTS, 300 TAKES TO END OF DAY
203       225   IF(X(1,M+1) - X(1,M))211,301,212
204       211   XIN = X(1,M) - 360.0
205             C: 301 TAKES OUT FOR INVALID DATA
206             GO TO 213
207       212   XIN = X(1,M)
208             C: TEST FOR 0 TIME PERIODS AND SET INCREMENT TO 0
209       213   GO TO(220,221),JC
210       220   GO TO(214,215,216,217,217),M
211       214   IF(JTIME - 0000)221,218,215
212       215   IF(JTIME - 0600)221,218,216
213       216   IF(JTIME - 1200)221,218,217
214       217   IF(JTIME - 1800)221,218,219
215       219   IF(JTIME - 2400)221,218,300
216       218   XINCG = 0.0
217             XINCD = 0.0
218             JC = JC + 1
219             GO TO 223
220       221   GO TO(233,222),JD
221       233   XINCG = (X(1,M+1) - XIN)*INTER/360.0
222             XINCD = (X(2,M+1) - X(2,M))*INTER/360.0
223             JD = JD + 1
224       222   X(1,M) = X(1,M) + XINCG
225             IF(X(1,M)-360.0)235,236,236
226       236   X(1,M) = X(1,M) - 360.0
227       235   X(2,M) = X(2,M) + XINCD
228             C: INCREMENTS GHA AND DEC AND RETURNS TO TEST RANGE
229             C: OUT OF INCREMENTS LOOP. CALC AZ AND EL
230       223   PSIA = X(1,M)
231             IF(BLONG - 180.0)226,227,227
232       226   IF(BLONG - X(1,M))228,227,227
233       228   PSIA = X(1,M) - 360.0
234             GO TO 229
235       227   IF((BLONG + 180.0).GT.360.0)GO TO 230
236             GO TO 229
237       230   PSIA = X(1,M) + 360.0
238       229   PSI = BLONG - PSIA
239             C: ABOVE TESTS FOR LONGS UNDER 100 AND GHA NR 360
240             C: CONVERT ANGLES TO RADIANS
241             PSI = PSI*CONVT
242             CXD = X(2,M)*CONVT
243             CSIND = SIN(CXD)
244             CCOSD = COS(CXD)
245             CCOSP = COS(PSI)
246             SEL = CCOSL*CCOSP*CCOSD + CSIND*CSINL
247             C: TAKES OUT FOR NEGATIVE EL ANGLES
248             IF(SEL)203,260,260
249       260   IF(SEL - 1.0)243,242,243
>
```

```
250      242    CEL = 0.0
251             TEL = 9.999999E36
252             GO TO 244
253      243    CEL = SQRT(1.0 - SEL**2)
254             TEL = SEL/CEL
255      244    CAZ1 = CSIND/(CEL*CCOSL) - CTANL*TEL
256             IF(CAZ1 - 1.0)246,245,246
257      245    CAZ = 0.0
258             TAZ = 9.999999E36
259             GO TO 247
260      246    CAZ = SQRT(1.0 - CAZ1**2)
261             TAZ = CAZ/CAZ1
262      247    ACAZ = ATAN(TAZ)
263             ASEL = ATAN(TEL)
264             ASEL = ASEL*CONVTR
265             ACAZ = ACAZ * CONVTR
266             GO TO(259,258),JE
267      259    JE = JE + 1
268             WRITE(1,128)
269             C: TEST FOR ANGLES ACROSS 360
270      258    IF((BLONG-180.0).LT.0.0.OR.(BLONG+180.0)
                              .GT.360.0)GO TO 256
272             GO TO 241
273             C: LONGITUDES 260 TO 360
274      256    IF((BLONG-180.0).LT.0.0)GO TO 257
275             IF(R2 .GT.R1.AND.X(1,M).GE.R2)GO TO 241
276             GO TO 240
277             C: LONGITUDES 000 TO 100
278      257    IF(X(1,M).LE.R1.AND.X(1,M).GE.(BLONG-180.0))
                             GO TO 241
280             GO TO 253
281      241    IF(BLONG - X(1,M))240,253,253
282      240    IF(CAZ1)250,250,251
283      250    ACAZ = 180.0 - ACAZ
284             GO TO 237
285      251    ACAZ = 360.0 - ACAZ
286             GO TO 237
287      253    IF(CAZ1)254,237,237
288      254    ACAZ = 180.0 + ACAZ
289      237    GO TO(231,232),JB
290      231    JB = JB + 1
291             WRITE(1,119)OBSER
292             WRITE(1,127)BLONG,BLAT
293             WRITE(1,129)DATE
294             WRITE(1,120)
295      232    WRITE(1,125)JTIME,X(1,M),X(2,M),ACAZ,ASEL
296             GO TO 224
297      301    WRITE(1,121)
298      300    WRITE(1,122)
299             WRITE(1,123)
300             ACCEPT K
301             IF(K - 1)302,201,201
302      302    WRITE(1,126)
303             WRITE(1,123)
304             ACCEPT K
305             IF(K - 1)303,200,200
306      303    STOP
307             END
>
```